

```

FINISH
/CLEAR
C*****
C***
C*** USO DI ELEMENTI PIANI
C***
C*** - LASTRA PIANA CON INTAGLIO IN "PLANE STRESS"
C*** - CORPO IN "PLANE STRAIN"
C*** - CORPO CILINDRICO CON INTAGLIO
C***
C*** PARAMETRI
C***
/PREP7
D=30 ! semi-larghezza / raggio nominale
L=90 ! semi-lunghezza
RR=10 ! raggio di raccordo
NC=12 ! n° di divisioni nella zona del raccordo
ESZ1=2 ! dimensioni elementi
PA=10 ! tensione nominale in senso assiale
C***
C*** AREE
C***
RECTNG,0,D,0,D
RECTNG,0,D,D,L
AGLUE,1,2 ! Unisce le due aree rettangolari
CYL4,D,0,RR
ASEL,,LOC,X,D-0.1,D+0.1 ! trova il numero del cerchio
*GET,NA,AREA,,NUM,MAX
ASEL,ALL
ASBA,1,NA
C***
C*** MATERIALE
C***
MP,EX,1,210000
C***
C*** VINCOLI DI SIMMETRIA
C***
LSEL,,LOC,Y,-1,0.001
DL,ALL,,SYMM
LSEL,ALL
LSEL,,LOC,X,-1,0.001
DL,ALL,,SYMM
LSEL,ALL
C***
C*** CARICHI
C***
LSEL,,LOC,Y,L-0.001,L+1
SFL,ALL,PRESS,-PA
C***
C*** MESH
C***
*ASK,TA,Tipo di analisi (<=0:plane stress, 1:assialsim., >=2:plane strain):,0
*IF,TA,LE,0,THEN
  K3=0
  /TITLE,STATO PIANO DI TENSIONE
*ELSEIF,TA,EQ,1
  K3=1
  /TITLE,STATO DI TENSIONE ASSIALSIMMETRICO
*ELSE
  K3=2
  /TITLE,STATO PIANO DI DEFORMAZIONE
*ENDIF

```

```
ET,1,42,,K3
ESIZE,ESZ1          ! dimensioni elementi
LSEL,,LOC,X,-0.001,0.001    ! concatena i due lati opposti al raccordo
LSEL,R,LOC,Y,-1,D-0.01     ! in modo che l'area abbia 4 lati
*GET,NL1,LINE,,NUM,MAX
LSEL,,LOC,Y,D-0.001,D+0.01
*GET,NL2,LINE,,NUM,MAX
LSEL,ALL
LCCAT,NL1,NL2
MSHAPE,0           ! parametri per il mesh "mapped"
MSHKEY,1
AMESH,ALL
C***
C*** SOLUZIONE
C***
FINISH
/SOLU
SOLVE
FINISH
C***
C*** POST-PROCESSING
C***
/POST1
SET
PLNSTR,S,Y
```